

## R17 Tool Changer instructions.

**Please read this before proceeding with the R17 manual.**

### Overview

Some ROBOFORTH 'code' is supplied with the system, see later.


The pneumatic devices comprise a pneumatic gripper and/or a vacuum pickup and/or an electric gripper and a tool changer system. The robot can pick up either the vacuum pickup or the gripper with the tool changer. The tool not in use is parked on a tool stand.

To operate the tool changer compressed air is required to unlock and release the tool. With no air to the changer a spring holds the tool in place but locking pressure is also used to ensure a positive lock against the electrical contacts.

### Pneumatics

#### **THIS ITEM IS MOUNTED ON PLASTIC BASE AND HAS TEMPORARY CONNECTIONS FOR TESTING ONLY**

The pneumatics circuit comprise 4 valves (or switches), a Venturi to produce vacuum, and vacuum sensor. See diagram1 below. Connect to the robot base as indicated in diagram2.



The controller has an output port PA with 8 bits, 3 of which control the pneumatic valves. PA 2 switches air either to the vacuum pickup or the gripper. If air is going to the vacuum pickup then PA 0 switches that on or off. If air is selected to go to the gripper then PA0 switches between open and closed.

To unlock the tool changer operate PA bit 3 with PA 3 ON. To lock again use PA 3 OFF.

To operate the vacuum simply use GRIPPER ON after setting PA 2 . To release vacuum use GRIPPER OFF. Note that while GRIPped air is continually flowing through the venturi so you will require a decent air supply.

To close the gripper first use PA 2 ON to switch air from vacuum to grip then use GRIP. Use UNGRIP to open the gripper. Note that the gripper must be held by the changer before operating PA 2 or air will escape through the tool changer.

All this is handled by the sample code supplied.

Words in the code:

LOCK – locks the changer

UNLOCK – unlocks the changer – make sure it is unlocked before you mate the 2 parts.

GVALVE – OFF for vacuum ON for pneumatic gripper.

### **Grip check (if fitted) or Vacuum check**

If the vacuum cups are holding something there will be a relatively high vacuum which will operate the vacuum sensor. This connects to the controller via the input port PB bit 6 which can be read with PB 6 BIT? – see sample code.

If you have a gripper sensor this is a small sensor in the side of the gripper which connects back to the controller through the tool changer and the robot to PB bit 5 and can be tested with PB 5 BIT?. The idea is that you slide this sensor along the slot to such a position that you get a true when the gripper is open OR when closed on an object, but you get a false if the gripper closes completely i.e. is gripping nothing.

You can monitor both these sensors with

PP

which reads in the bit pattern from port PB and displays it as a line of 1 and 0s e.g.

11111111

right to left is most significant to least significant. When you operate the gripper sensor

11011111

When you operate the vacuum sensor you will see

10111111

Press escape to exit PP

Adjust the gripper sensor so it will differentiate between the object and no object and lock in place with the small screw. The vacuum sensor is also adjustable using a 1mm metric Allen key.

There is a test word VP

When you enter VP the vacuum is turned on and PP is entered so you can adjust the vacuum sensor. Press esc and the vacuum turns off.

### How to program the tool change.

Go to the safe position

READY or for a 6 axis robot READY2

The robot will end up at a position with the hand center level with the shoulder pivot, pointing down and with 350mm between these centers.

Once you have familiarized yourself with the use of projects you can load the sample project supplied on the CD. In project R17TOOLCHANGE.RUN (see below) the positions where the tools are parked are predefined as positions in 3 routes, GPATH (for the gripper) and VPATH and VPOUT (for the vacuum pickup).

Starting with the gripper, in RobWin select and open GPATH.

Next click the J (jog) button and move the robot over the rack and down into the first tool using the teach pad.


Once closed with the tool (a small gap is tolerated) exit the jog mode, click line 4 which is the final lock position in the route. Click 'set to here'.

Now move up by 20mm, select line 3 and 'set to here'

Move up another 20mm, select line 2 and 'set to here'

Move up until well clear of the rack (say 50mm), select line 1 and 'set to here'.

See pictures below



GETGRIP will get the gripper.

PUTGRIP will put it back. It works by using RETRACE on the route GPATH. See the sample code below.

The vacuum pickup has a similar route VPATH but it is more complicated because you can not pull the vacuum pickup straight up so you can't use RETRACE on VPATH. So a second route is required to pull the pickup out of the rack sideways. This is named VPOUT in the example project.

Select and open VPATH

Move the robot across and carefully down into the tool with the Jog function.

Repeat the same procedure as for the gripper and GPATH but using VPATH.

Run it i.e. VPATH RUN so the robot is now in the tool.

Open VPOUT for editing.

Select line 1 and click set to here since line 1 of VPOUT is the same as the last line of VPATH.

Using the teach pad go up 10mm, highlight line 2 and click set to here.

Start moving sideways and out of the rack. You will have to use some J5 (roll – ) because the pillars in the pickup and not at 90 degrees to the locator. Once out of the rack select line 3 and set to here. Then move up a long distance and select line 4 and set to here.



VPOUT line 1



line 2



line 3



line 4

On RobWin use project, save.

You should now be able to put the vacuum pickup back with PUTVAC which RETRACES VPOUT, UNLOCKS then RETRACES VPATH.

If you want to create these routes from scratch in a new project then what you do is create new Cartesian routes called GPATH, VPATH and VPOUT (or anything you prefer to name them). Starting with GPATH teach down into the tool until you have a good fit. At the pickup position use insert position to learn that. Then progressively move upwards on Z using 'insert position' each time so that the pickup position is always the last line. Then when you RUN the route it starts well clear and finishes at the pickup position. RETRACE moves it back out again.

Open toolchanger.ed2 as a file, copy all the text and insert it into your new ed2 text window. After creating and teaching VPATH and/or GPATH you will need to click the green down-arrow to get the definitions of LOCK and UNLOCK, PUTVAC etc. in your new project.

Here is a help video: <https://www.youtube.com/watch?v=nI9wHfUC-KY>

If your vacuum pickup is larger than the one in the video you will need to create 2 routes as described above.

### SAMPLE CODE

( STARTER PROJECT FOR TOOL CHANGING - PNEU PLUS VAC

### DECIMAL

```
: CHANGER PA 3 ;  
: LOCK CHANGER OFF 200 MSECs ;  
: UNLOCK CHANGER ON 200 MSECs ;  
: GVALVE PA 2 ; ( OFF = VAC, ON = GRIPPER
```

```
: SLOW 1000 SPEED ! ;
```

```
: PUTGRIP  
GRIPPER OFF  
GVALVE OFF  
NORMAL  
GPATH SMOOTH RUN  
UNLOCK  
RETRACE  
;
```

```
: GETGRIP  
UNLOCK  
NORMAL  
GPATH SMOOTH RUN  
LOCK  
RETRACE  
GVALVE ON  
;
```

```
: PUTVAC  
GRIPPER OFF  
NORMAL  
VPOUT SMOOTH RETRACE RESTORE  
UNLOCK  
VPATH SMOOTH RETRACE RESTORE  
;  
: GETVAC  
UNLOCK  
NORMAL  
VPATH SMOOTH RUN RESTORE  
LOCK  
VPOUT SMOOTH RUN RESTORE  
GVALVE OFF  
;
```

```
: VACCHECK  
PB 6 BIT? IF ( VAC SENSE NOT LOW  
CR ." VACUUM PICK FAIL"  
0 OBJECT-HELD !  
PA 0 OFF ( TO SAVE AIR  
15 ABORT  
THEN  
;
```



---

```
: VP GRIP PP UNGRIP ; ( TO ADJUST SENSOR
```

```
: TEST
UNLOCK
  GETVAC
  PUTVAC
500 MSECs
  GETGRIP
  PUTGRIP
;
```

The next 2 defined words allow you to lock and unlock the changer using the FN key on the teach pad or the Function button on the Android teach console.

```
: LOCK-TOG
  PA 3 BIT? IF LOCK ELSE UNLOCK THEN
;
SET FN LOCK-TOG ( include this in your initializing words
```

**When using an electric gripper** you must block the air when picking the electric gripper. This is done with PA 2. If there is no vacuum pickup then the unused port will be blocked. Close that valve to direct the pressure to the blocked port.

Then you will need to tell the system which type of GRIP. This is done by changing the value of GTYPE to 0 for pneumatic or 1 for electric:

```
1 GTYPE ! ( SELECTS ELECTRIC GRIPPER
0 GTYPE ! ( SELECTS PNEUMATIC GRIPPER
```

Suggested words:

```
: CHANGER PA 3 ;
: GVALVE PA 2 ; ( OFF = ELEC, ON = PNEU
: LOCK CHANGER OFF 200 MSECs GVALVE ON ;
: UNLOCK GVALVE OFF CHANGER ON 200 MSECs ;
: GRIP
GTYPE C@ IF ( ELECTRIC GRIPPER
  GVALVE OFF ( BLOCK THE AIR
ELSE ( PNEUMATIC GRIPPER
  GVALVE ON ( ENABLE THE AIR
THEN
100 MSECs
GRIP
;
```

The electric gripper will also require 2 routes as described for the vacuum pickup above, say EGPATH and EGOUT. You can see how this works out here:

<https://www.youtube.com/watch?v=PXUCqtbAC1I>